```
Clear["Global`*"]
$Line = 0;
```

CFT and Gravity 2017

Week 3 Exercise 2.2.1

Define $g_{\mu\nu}$, a scalar product · and rules for contractions

```
Attributes[CenterDot] = {Orderless};
(a_ + b_) · c_ := a · c + b · c
a_ · (α_ b_) := α (a · b) /; NumberQ[α]
Attributes[g] = Orderless;
dot := ((Expand[#] //. {
        z_[μ_] y_[μ_] :→ z · y, z_[μ_]^2 :→ z · z, g[μ_, ν_] z_[ν_] :→ z[μ], g[μ_, α_]
          g[α_, ν_] :→ g[μ, ν] , g[μ_, ν_]^2 :→ d, g[μ_, μ_] :→ d}) // Simplify) &
```

Warning: it does not handle mistakes in the repeated indices: see first example below. So please contract indices properly in the following.

```
x[μ] x[α] y[μ] x[μ] // dot
x[μ] x[α] y[μ] x[α] // dot
x[μ] (x[μ] + y[μ]) // dot
x[μ] (x[ν] + y[ν]) g[μ, ν] // dot
g[μ, ν] g[ν, μ] // dot
```

```
x · x x[α] y[μ]
```

```
x · x x · y
```

```
x · x + x · y
```

```
x · x + x · y
```

```
d
```

Define ∂_μ

der[x_][μ_][expr_] takes three arguments: the point x, the direction μ and the expression to be derived. We use the command SeriesCoefficient that knows already about linearity and Leibnitz rule:

```
der[x_][μ_][A_] := (A /. {
      x[α_] :→ x[α] + ϵ g[μ, α], x · x :→ x · x + 2 ϵ x[μ], x · y_ :→ x · y + ϵ y[μ]}) //
   SeriesCoefficient[#, {ϵ, 0, 1}] &
```

```
der[x][μ][x · x]
der[x][μ][x · y]
der[x][μ][f[x · y]]
```

```
2 x[μ]
```

```
y[μ]
```

```
y[μ] f′[x · y]
```

- **Define the generators of the conformal group at point x:**

```
P[μ_][A_] := -ⅈ der[x][μ][A]
M[μ_, ν_][A_] := -ⅈ (x[μ] der[x][ν][A] - x[ν] der[x][μ][A])
Di[A_] := -Module[{λ}, x[λ] der[x][λ][A]]
Ks[μ_][A_] := 2 ⅈ x[μ] Module[{λ}, x[λ] der[x][λ][A]] - ⅈ x · x der[x][μ][A]
```

- **Compute the algebra by applying to a generic function.**

We must choose the function to depend on the full vector x[*μ*] rather than on x^2, otherwise the rotation generator acts trivially.

```
(Di[P[μ][#]] - P[μ][Di[#]] == P[μ][#]) &[f[x · k]] // dot
(Di[Ks[μ][#]] - Ks[μ][Di[#]] == -Ks[μ][#]) &[f[x · y, x · z]] // dot
(Ks[μ][P[ν][#]] - P[ν][Ks[μ][#]] == 2 g[μ, ν] Di[#] - 2 ⅈ M[μ, ν][#]) &[f[x · y]] //
 dot
(M[μ, ν][P[α][#]] - P[α][M[μ, ν][#]] == ⅈ (g[μ, α] P[ν][#] - g[ν, α] P[μ][#])) &[
  f[x · y]] // dot
(M[μ, ν][Ks[α][#]] - Ks[α][M[μ, ν][#]] == ⅈ (g[μ, α] Ks[ν][#] - g[ν, α] Ks[μ][#])) &[
  f[x · y]] // dot
(M[α, β][M[μ, ν][#]] - M[μ, ν][M[α, β][#]] ==
     ⅈ (g[α, μ] M[β, ν][#] + g[β, ν] M[α, μ][#] -
         g[α, ν] M[β, μ][#] - g[β, μ] M[α, ν][#])) &[f[x · y]] // dot
```

```
True
```

```
True
```

```
True
```

```
True
```

```
True
```

```
True
```

Week 3 exercise 2.3.6

- **a)**

The expression proposed in the exercised is clearly covariant under the Poincaré group, therefore we only need to check the behavior under inversion. Instead of using brute force, let’s follow the hints given in the text.

Check how distances transform under inversions:

```
(x1[μ]/(x1 . x1) - x2[μ]/(x2 . x2))^2 == (x1[μ] - x2[μ])^2/(x1 . x1 x2 . x2) // dot
```

```
True
```

Check the transformation properties of V^μ:

```
V[μ_] = (x₁[μ] - x₃[μ])/dist[1, 3] - (x₂[μ] - x₃[μ])/dist[2, 3]
```

```
(x₁[μ] - x₃[μ])/dist[1, 3] - (x₂[μ] - x₃[μ])/dist[2, 3]
```

Btw, the Jacobian of the inversion is closely related to the inversion matrix:

```
Inv[x_][μ_, ν_] := g[μ, ν] - 2 (x[μ] x[ν])/(x . x)
```

```
der[x][μ][x[ν]/(x . x)] == 1/(x . x) Inv[x][μ, ν] // Simplify
```

```
True
```

This checks the transformation rule of V^μ

```
V[μ] /. {x_i_[μ] -> x_i[μ]/(x_i · x_i), dist[i_, j_] -> dist[i, j]/(x_i · x_i x_j · x_j)} // Expand
1/(x_3 · x_3) Inv[x_3][ν, μ] % // dot // Expand
(% == V[ν]) /. dist[i_, j_] → (x_i - x_j) · (x_i - x_j) // Simplify
```

$$\frac{x_3 \cdot x_3\, x_1[\mu]}{\text{dist}[1,3]} - \frac{x_3 \cdot x_3\, x_2[\mu]}{\text{dist}[2,3]} - \frac{x_1 \cdot x_1\, x_3[\mu]}{\text{dist}[1,3]} + \frac{x_2 \cdot x_2\, x_3[\mu]}{\text{dist}[2,3]}$$

$$\frac{x_1[\nu]}{\text{dist}[1,3]} - \frac{x_2[\nu]}{\text{dist}[2,3]} + \frac{x_1 \cdot x_1\, x_3[\nu]}{x_3 \cdot x_3\, \text{dist}[1,3]} - \frac{2\, x_1 \cdot x_3\, x_3[\nu]}{x_3 \cdot x_3\, \text{dist}[1,3]} - \frac{x_2 \cdot x_2\, x_3[\nu]}{x_3 \cdot x_3\, \text{dist}[2,3]} + \frac{2\, x_2 \cdot x_3\, x_3[\nu]}{x_3 \cdot x_3\, \text{dist}[2,3]}$$

```
True
```

This takes care of the tensor part in the transformation law of j_μ
All what is left is to match scaling dimensions.
The inversion is an orthogonal matrix:

```
Inv[x][μ, ν] Inv[x][ν, ρ] // dot
```

```
g[μ, ρ]
```

so it has eigenvalues ±1. This means
$\left| \frac{\partial x'}{\partial x} \right| = (x^2)^{-d}$.

Define the prefactors of V^μ in the correlator:

```
pref = dist[1, 2]^α12 dist[1, 3]^α13 dist[2, 3]^α23;
```

pref must obey the following relation:
pref = pref' $(x1^2)^{-\Delta 1} (x2^2)^{-\Delta 2} (x3^2)^{-(\Delta 3-1)}$

```
pref /. dist[i_, j_] -> dist[i, j]/(x_i · x_i x_j · x_j)
% (x_1 · x_1)^-Δ1 (x_2 · x_2)^-Δ2 (x_3 · x_3)^-(Δ-1);
% / pref // PowerExpand
Solve[Cases[%, (x_i_ · x_i_)^a_ → a] == 0, {α12, α13, α23}]
```

$$\left(\frac{\text{dist}[1,2]}{x_1 \cdot x_1\, x_2 \cdot x_2}\right)^{\alpha 12} \left(\frac{\text{dist}[1,3]}{x_1 \cdot x_1\, x_3 \cdot x_3}\right)^{\alpha 13} \left(\frac{\text{dist}[2,3]}{x_2 \cdot x_2\, x_3 \cdot x_3}\right)^{\alpha 23}$$

$$(x_1 \cdot x_1)^{-\alpha 12-\alpha 13-\Delta 1}\, (x_2 \cdot x_2)^{-\alpha 12-\alpha 23-\Delta 2}\, (x_3 \cdot x_3)^{1-\alpha 13-\alpha 23-\Delta}$$

$$\left\{\left\{\alpha 12 \to \frac{1}{2}\left(-1+\Delta-\Delta 1-\Delta 2\right),\ \alpha 13 \to \frac{1}{2}\left(1-\Delta-\Delta 1+\Delta 2\right),\ \alpha 23 \to \frac{1}{2}\left(1-\Delta+\Delta 1-\Delta 2\right)\right\}\right\}$$

this coincides with the proposed form of the correlator.

- **b)**

The product $V^\mu V^\nu$ clearly respects the tensor transformation rule, and so does $g_{\mu\nu}$. On the other hand $V_\alpha V^\alpha$ is invariant. The combination
$H_{\mu\nu} = V_\mu V_\nu - \frac{1}{d} V^\alpha V_\alpha g_{\mu\nu}$
therefore also respects the rule, and the coefficient is fixed by requiring $H^\mu{}_\mu = 0$.
The only thing to be checked is again the scaling of the prefactor, and the check is almost the same as in the spin 1 case.

- **Unicity**

So far we verified that a certain tensor structure is compatible with the transformation property of the correlator. We can actually argue that there can be only one.

The argument goes as follows:

1. Use conformal transformations to send O_1 to the origin and O_2 to infinity.
2. Now the correlator only depends on $x_3{}^\mu$, and there is a unique traceless symmetric tensor built out of a single vector.
3. Suppose that there were two tensor structures that become degenerate when $x_1 \to 0$ and $x_2 \to \infty$. This is inconsistent because the conformal transformations are invertible.

- **d)**

Let's build a function that applies Wick theorem (an overkill for this exercise, but useful in general).
First we implement point splitting for fields evaluated at the same point:

```
split[a_] := List @@ (a /. ϕ[b_]^n_ :> Times @@ Table[ϕ[b[i]], {i, 1, n}])
```

Then we define the wick function to take a list of fields.

```
wick[{}] := 1;
wick[{num_}] := num /; NumberQ[num]
wick[{a_}] := 0
wick[{num_, a__}] := num wick[{a}] /; NumberQ[num]
wick[{a_, b_}] := 〈a b〉
wick[a_ ] := Module[{aux = Delete[a, 1]},
  Sum[〈a[[1]] aux[[j]]〉 wick[Delete[aux, j]], {j, 1, Length[aux]}]]
```

For instance:

```
wick[{ϕ[x1], ϕ[x2], ϕ[x3], ϕ[x4]}]
```

```
〈ϕ[x2] ϕ[x3]〉 〈ϕ[x1] ϕ[x4]〉 + 〈ϕ[x1] ϕ[x3]〉 〈ϕ[x2] ϕ[x4]〉 + 〈ϕ[x1] ϕ[x2]〉 〈ϕ[x3] ϕ[x4]〉
```

We define $T_{\mu\nu}$ as the sum of two differential operators which act on fields.
$\partial_\mu \phi \partial_\nu \phi$ is defined via point splitting, while the second piece is a differential operator directly acting on ϕ^2

```
t1[x_, y_][μ_, ν_] = der[x][μ][der[y][ν][#]] &;
t2[x_][μ_, ν_] =
   -1/(4 (d - 1)) ((d - 2) der[x][μ][der[x][ν][#]] + g[μ, ν] der[x][λ][der[x][λ][#]]) &;
```

The correlator is obtained by applying these differential operator to the appropriate scalar correlators:

t1: notice that we need to subtract contractions between ϕ[x] and ϕ[y] since in $T_{\mu\nu}$ these fields are normal ordered and evaluated in the same point:

```
split[ϕ[x1] ϕ[x2] ϕ[xs] ϕ[ys]];
wick[%] /. ⟨ϕ[xs] ϕ[ys]⟩ → 0 /. ⟨ϕ[a_] ϕ[b_]⟩ → norm / ((a - b) · (a - b))^((d-2)/2)
t1piece = t1[xs, ys][μ, ν][%] /. {xs → x3, ys → x3};
```

$$\mathrm{norm}^2\,(x_2\cdot x_2-2\,x_2\cdot x_s+x_s\cdot x_s)^{\frac{2-d}{2}}\,(x_1\cdot x_1-2\,x_1\cdot y_s+y_s\cdot y_s)^{\frac{2-d}{2}}+\mathrm{norm}^2\,(x_1\cdot x_1-2\,x_1\cdot x_s+x_s\cdot x_s)^{\frac{2-d}{2}}\,(x_2\cdot x_2-2\,x_2\cdot y_s+y_s\cdot y_s)^{\frac{2-d}{2}}$$

```
split[ϕ[x1] ϕ[x2] ϕ[x]^2]
wick[%] /. ⟨ϕ[x[1]] ϕ[x[2]]⟩ → 0 /. {x[1] → x3, x[2] → x3} /.
  ⟨ϕ[a_] ϕ[b_]⟩ → norm / ((a - b) · (a - b))^((d-2)/2)
t2piece = t2[x3][μ, ν][%] // dot;
```

```
{ϕ[x1], ϕ[x2], ϕ[x[1]], ϕ[x[2]]}
```

$$2\,\mathrm{norm}^2\,(x_1\cdot x_1-2\,x_1\cdot x_3+x_3\cdot x_3)^{\frac{2-d}{2}}\,(x_2\cdot x_2-2\,x_2\cdot x_3+x_3\cdot x_3)^{\frac{2-d}{2}}$$

```
tcorr = t1piece + t2piece;
```

Now we compare this result with the prediction from conformal invariance.

```
(V[μ] V[ν] - 1/d V[α] V[α] g[μ, ν] // dot);
c12T % / (dist[1, 2]^((Δ1+Δ2-Δ+2)/2) dist[1, 3]^((Δ1+Δ-Δ2-2)/2) dist[2, 3]^((Δ2+Δ-Δ1-2)/2)) /.
  dist[i_, j_] → (xi - xj) · (xi - xj);
zero = % - tcorr // Expand //
   Collect[#, {g[μ, ν], x1[μ] x1[ν], x2[μ] x2[ν], x3[μ] x3[ν], x1[μ] x2[ν],
      x1[μ] x3[ν], x2[μ] x3[ν], x1[ν] x1[μ], x2[ν] x2[μ], x3[ν] x3[μ], x1[ν] x2[μ],
      x1[ν] x3[μ], x2[ν] x3[μ]}, f[#] &] & // Cases[#, f[a_] → a, ∞] &
```

```
{- 1/(-1+d) 2 norm^2 (… 1 …)^2 (… 1 …) (… 1 …)^(-1-d/2)
   (x2 · x2 - 2 x2 · (… 1 …) + x3 · x3)^(-1-d/2) + 2 (… 5 …) (… 1 …)^(… 1 …) / (-1+d) -
   (… 1 …) / (2 (… 1 …)) + (… 148 …) + (… 1 …) / d - x2 · x2 (… 4 …) c12T / d , (… 8 …) , (… 1 …)}
large output   show less   show more   show all   set size limit...
```

What we did here is to subtract the expected form to the computed one: so zero should be zero.
Each tensor structure gives an independent equation (at least in principle), so we built a list of the

coefficients of all the tensor structures.

The result is quite complicated, so we can introduce a bit of notation to simplify it:

```
zerosimp =
  zero /. x_{i_} · x_{j_} → -1/2 (dist[i, j] - x_i · x_i - x_j · x_j) /. dist[i_, i_] → 0 // Simplify
```

$$\Big\{\frac{1}{2(-1+d)\,d}\,\mathrm{dist}[1,2]^{\frac{1}{2}(-\Delta1-\Delta2)}\,\mathrm{dist}[1,3]^{\frac{1}{2}(-d-\Delta-\Delta1)}\,\mathrm{dist}[2,3]^{\frac{1}{2}(-d-\Delta-\Delta2)}\Big(-(-2+d)^2\,d\,\mathrm{norm}^2\,\mathrm{dist}[1,2]^{\frac{1}{2}(2+\Delta1+\Delta2)}\,\mathrm{dist}[1,3]^{\frac{\Delta+\Delta1}{2}}\,\mathrm{dist}[2,3]^{\frac{\Delta+\Delta2}{2}}-2(-1+d)\,\mathrm{dist}[1,2]^{\Delta/2}\,\mathrm{dist}[1,3]^{\frac{d+\Delta2}{2}}\,\mathrm{dist}[2,3]^{\frac{d+\Delta1}{2}}\,c_{12\,T}\Big),$$

$$\frac{1}{2(-1+d)}\,\mathrm{dist}[1,2]^{\frac{1}{2}(-2-\Delta1-\Delta2)}\,\mathrm{dist}[1,3]^{\frac{1}{2}(-2-d-\Delta-\Delta1)}\,\mathrm{dist}[2,3]^{\frac{1}{2}(2-d-\Delta-\Delta2)}\Big((-2+d)^2\,d\,\mathrm{norm}^2\,\mathrm{dist}[1,2]^{\frac{1}{2}(2+\Delta1+\Delta2)}\,\mathrm{dist}[1,3]^{\frac{\Delta+\Delta1}{2}}\,\mathrm{dist}[2,3]^{\frac{\Delta+\Delta2}{2}}+2(-1+d)\,\mathrm{dist}[1,2]^{\Delta/2}\,\mathrm{dist}[1,3]^{\frac{d+\Delta2}{2}}\,\mathrm{dist}[2,3]^{\frac{d+\Delta1}{2}}\,c_{12\,T}\Big),$$

$$\frac{1}{2(-1+d)}\,\mathrm{dist}[1,2]^{\frac{1}{2}(-2-\Delta1-\Delta2)}\,\mathrm{dist}[1,3]^{\frac{1}{2}(-d-\Delta-\Delta1)}\,\mathrm{dist}[2,3]^{\frac{1}{2}(-d-\Delta-\Delta2)}\Big(-(-2+d)^2\,d\,\mathrm{norm}^2\,\mathrm{dist}[1,2]^{\frac{1}{2}(2+\Delta1+\Delta2)}\,\mathrm{dist}[1,3]^{\frac{\Delta+\Delta1}{2}}\,\mathrm{dist}[2,3]^{\frac{\Delta+\Delta2}{2}}-2(-1+d)\,\mathrm{dist}[1,2]^{\Delta/2}\,\mathrm{dist}[1,3]^{\frac{d+\Delta2}{2}}\,\mathrm{dist}[2,3]^{\frac{d+\Delta1}{2}}\,c_{12\,T}\Big),$$

$$\frac{1}{2(-1+d)}\,\mathrm{dist}[1,2]^{\frac{1}{2}(-2-\Delta1-\Delta2)}\,\mathrm{dist}[1,3]^{\frac{1}{2}(-d-\Delta-\Delta1)}\,\mathrm{dist}[2,3]^{\frac{1}{2}(-d-\Delta-\Delta2)}\Big(-(-2+d)^2\,d\,\mathrm{norm}^2\,\mathrm{dist}[1,2]^{\frac{1}{2}(2+\Delta1+\Delta2)}\,\mathrm{dist}[1,3]^{\frac{\Delta+\Delta1}{2}}\,\mathrm{dist}[2,3]^{\frac{\Delta+\Delta2}{2}}-2(-1+d)\,\mathrm{dist}[1,2]^{\Delta/2}\,\mathrm{dist}[1,3]^{\frac{d+\Delta2}{2}}\,\mathrm{dist}[2,3]^{\frac{d+\Delta1}{2}}\,c_{12\,T}\Big),$$

$$\frac{1}{2(-1+d)}\,\mathrm{dist}[1,2]^{\frac{1}{2}(-2-\Delta1-\Delta2)}\,\mathrm{dist}[1,3]^{\frac{1}{2}(2-d-\Delta-\Delta1)}\,\mathrm{dist}[2,3]^{\frac{1}{2}(-2-d-\Delta-\Delta2)}\Big((-2+d)^2\,d\,\mathrm{norm}^2\,\mathrm{dist}[1,2]^{\frac{1}{2}(2+\Delta1+\Delta2)}\,\mathrm{dist}[1,3]^{\frac{\Delta+\Delta1}{2}}\,\mathrm{dist}[2,3]^{\frac{\Delta+\Delta2}{2}}+2(-1+d)\,\mathrm{dist}[1,2]^{\Delta/2}\,\mathrm{dist}[1,3]^{\frac{d+\Delta2}{2}}\,\mathrm{dist}[2,3]^{\frac{d+\Delta1}{2}}\,c_{12\,T}\Big),$$

$$\frac{1}{2(-1+d)}\,\mathrm{dist}[1,2]^{\frac{1}{2}(-2-\Delta1-\Delta2)}\,\mathrm{dist}[1,3]^{\frac{1}{2}(-2-d-\Delta-\Delta1)}\big(\mathrm{dist}[1,3]-\mathrm{dist}[2,3]\big)\,\mathrm{dist}[2,3]^{\frac{1}{2}(-d-\Delta-\Delta2)}\Big((-2+d)^2\,d\,\mathrm{norm}^2\,\mathrm{dist}[1,2]^{\frac{1}{2}(2+\Delta1+\Delta2)}\,\mathrm{dist}[1,3]^{\frac{\Delta+\Delta1}{2}}\,\mathrm{dist}[2,3]^{\frac{\Delta+\Delta2}{2}}+2(-1+d)\,\mathrm{dist}[1,2]^{\Delta/2}\,\mathrm{dist}[1,3]^{\frac{d+\Delta2}{2}}\,\mathrm{dist}[2,3]^{\frac{d+\Delta1}{2}}\,c_{12\,T}\Big),$$

$$-\frac{1}{2(-1+d)}\,\mathrm{dist}[1,2]^{\frac{1}{2}(-2-\Delta1-\Delta2)}\,\mathrm{dist}[1,3]^{\frac{1}{2}(-d-\Delta-\Delta1)}\big(\mathrm{dist}[1,3]-\mathrm{dist}[2,3]\big)\,\mathrm{dist}[2,3]^{\frac{1}{2}(-2-d-\Delta-\Delta2)}\Big((-2+d)^2\,d\,\mathrm{norm}^2\,\mathrm{dist}[1,2]^{\frac{1}{2}(2+\Delta1+\Delta2)}\,\mathrm{dist}[1,3]^{\frac{\Delta+\Delta1}{2}}\,\mathrm{dist}[2,3]^{\frac{\Delta+\Delta2}{2}}+2(-1+d)\,\mathrm{dist}[1,2]^{\Delta/2}\,\mathrm{dist}[1,3]^{\frac{d+\Delta2}{2}}\,\mathrm{dist}[2,3]^{\frac{d+\Delta1}{2}}\,c_{12\,T}\Big),$$

$$\frac{1}{2(-1+d)}\,\mathrm{dist}[1,2]^{\frac{1}{2}(-2-\Delta1-\Delta2)}\,\mathrm{dist}[1,3]^{\frac{1}{2}(-2-d-\Delta-\Delta1)}$$

$$\left(\text{dist}[1,3]-\text{dist}[2,3]\right)\text{dist}[2,3]^{\frac{1}{2}(-d-\Delta-\Delta 2)}$$
$$\left(\left(-2+d\right)^2 d\,\text{norm}^2\,\text{dist}[1,2]^{\frac{1}{2}(2+\Delta 1+\Delta 2)}\,\text{dist}[1,3]^{\frac{\Delta+\Delta 1}{2}}\,\text{dist}[2,3]^{\frac{\Delta+\Delta 2}{2}}+\right.$$
$$\left.2\left(-1+d\right)\text{dist}[1,2]^{\Delta/2}\,\text{dist}[1,3]^{\frac{d+\Delta 2}{2}}\,\text{dist}[2,3]^{\frac{d+\Delta 1}{2}}\,c_{12\,T}\right),$$
$$-\frac{1}{2\left(-1+d\right)}\text{dist}[1,2]^{\frac{1}{2}(-2-\Delta 1-\Delta 2)}\,\text{dist}[1,3]^{\frac{1}{2}(-d-\Delta-\Delta 1)}\left(\text{dist}[1,3]-\text{dist}[2,3]\right)$$
$$\text{dist}[2,3]^{\frac{1}{2}(-2-d-\Delta-\Delta 2)}\left(\left(-2+d\right)^2 d\,\text{norm}^2\,\text{dist}[1,2]^{\frac{1}{2}(2+\Delta 1+\Delta 2)}\,\text{dist}[1,3]^{\frac{\Delta+\Delta 1}{2}}\right.$$
$$\left.\text{dist}[2,3]^{\frac{\Delta+\Delta 2}{2}}+2\left(-1+d\right)\text{dist}[1,2]^{\Delta/2}\,\text{dist}[1,3]^{\frac{d+\Delta 2}{2}}\,\text{dist}[2,3]^{\frac{d+\Delta 1}{2}}\,c_{12\,T}\right),$$
$$\frac{1}{2\left(-1+d\right)}\text{dist}[1,2]^{\frac{1}{2}(-2-\Delta 1-\Delta 2)}\,\text{dist}[1,3]^{\frac{1}{2}(-2-d-\Delta-\Delta 1)}$$
$$\left(\text{dist}[1,3]-\text{dist}[2,3]\right)^2\text{dist}[2,3]^{\frac{1}{2}(-2-d-\Delta-\Delta 2)}$$
$$\left(\left(-2+d\right)^2 d\,\text{norm}^2\,\text{dist}[1,2]^{\frac{1}{2}(2+\Delta 1+\Delta 2)}\,\text{dist}[1,3]^{\frac{\Delta+\Delta 1}{2}}\,\text{dist}[2,3]^{\frac{\Delta+\Delta 2}{2}}+\right.$$
$$\left.2\left(-1+d\right)\text{dist}[1,2]^{\Delta/2}\,\text{dist}[1,3]^{\frac{d+\Delta 2}{2}}\,\text{dist}[2,3]^{\frac{d+\Delta 1}{2}}\,c_{12\,T}\right)\Big\}$$

Still long, but much better: in particular the first equation looks very promising. Let's isolate it and see how many parameters it determines:

```
zerosimp[[1, -1]] // Expand // Simplify
% // List @@ # & // #[[1]]/#[[2]] & // Cases[#, dist[i_, j_]^α_ → α == 0] &
dimensions = Solve[%, {Δ1, Δ2, Δ}][[1]]
%%% /. dimensions // Simplify
c12t = Solve[% == 0, c12T][[1]]
```

$$-\left(-2+d\right)^2 d\,\text{norm}^2\,\text{dist}[1,2]^{\frac{1}{2}(2+\Delta 1+\Delta 2)}\,\text{dist}[1,3]^{\frac{\Delta+\Delta 1}{2}}\,\text{dist}[2,3]^{\frac{\Delta+\Delta 2}{2}}-$$
$$2\left(-1+d\right)\text{dist}[1,2]^{\Delta/2}\,\text{dist}[1,3]^{\frac{d+\Delta 2}{2}}\,\text{dist}[2,3]^{\frac{d+\Delta 1}{2}}\,c_{12\,T}$$

$$\left\{-\frac{\Delta}{2}+\frac{1}{2}\left(2+\Delta 1+\Delta 2\right)=0,\ \frac{\Delta+\Delta 1}{2}+\frac{1}{2}\left(-d-\Delta 2\right)=0,\ \frac{1}{2}\left(-d-\Delta 1\right)+\frac{\Delta+\Delta 2}{2}=0\right\}$$

$$\left\{\Delta 1\to\frac{1}{2}\left(-2+d\right),\ \Delta 2\to\frac{1}{2}\left(-2+d\right),\ \Delta\to d\right\}$$

$$-\text{dist}[1,2]^{d/2}\,\text{dist}[1,3]^{-\frac{1}{2}+\frac{3d}{4}}\,\text{dist}[2,3]^{-\frac{1}{2}+\frac{3d}{4}}\left(\left(-2+d\right)^2 d\,\text{norm}^2+2\left(-1+d\right)c_{12\,T}\right)$$

$$\left\{c_{12\,T}\to-\frac{\left(-2+d\right)^2 d\,\text{norm}^2}{2\left(-1+d\right)}\right\}$$

So one equation already determines both the scaling dimensions and the coefficient of the three point function. All the others better be automatically satisfied:

```
zero /. dimensions /. c12t // Simplify
```

$$\{0,0,0,0,0,0,0,0,0,0\}$$